
Today’s Topics

• Link with Block A and Block B

• Motivation for Abstract Data Types

• Simple Graphical User Interfaces in Lazarus

An Easy Programming Problem: Windows 2001

Problem G, Prelims, Dutch Programming Championship NKP 1998

A Hacked Solution (with a mistake)

1 program Windows2001h; { Hacked version }
2 var r,n,m,x,y,j,k:integer;
3 w:array[1..10000] of record xl,xh,yl,yh:integer end;
4 begin
5 readln(r);
6 for r:=1 to r do begin readln(n);
7 for n:=1 to n do with w[n] do readln(xl,xh,yl,yh);
8 readln(m);
9 for m:=1 to m do begin readln(x,y);k:=1;j:=n+1;

10 while k<>j do with w[k] do
11 if (xl<=x)and(x<=xh)and(yl<=y)and(y<=yh) then j:=k
12 else k:=k+1;
13 if k<=n then writeln(k) else writeln(’Desktop’)
14 end end end.

A Monolithic Solution (using an array)

1 program Windows2001m1;
2 { Monolithic version using an array to store the windows }
3

4 var
5 r: Integer; { number of runs (input) }
6 i: Integer; { to count off the runs }
7 n: Integer; { number of windows (input) }
8 w: array [1..10000] of record
9 xl, yl, xh, yh: Integer; { coordinates of a window }

10 end; { w[1..n] = list of windows (input) }
11 j: Integer; { to count off the windows }
12 m: Integer; { number of mouse clicks (input) }
13 c: Integer; { to count off the mouse clicks }
14 x, y: Integer; { coordinates of a mouse click (input) }
15 k: Integer; { window number (output) }
16

17 begin
18 ReadLn(r) { number of runs }

A Monolithic Solution (using an array)

19 ; for i := 1 to r do begin
20

21 { Read and store the windows }
22 ReadLn(n) { number of windows }
23 ; for j := 1 to n do with w[j] do
24 ReadLn(xl, yl, xh, yh) { coordinates of the window }
25

26 { Read and process the mouse clicks }
27 ; ReadLn(m) { number of mouse clicks }
28 ; for c := 1 to m do begin
29 ReadLn(x, y) { coordinates of the mouse click }
30

31 { Find first window that contains (x,y)
32 using a Bounded Linear Search }
33 ; k := 1 ; j := n + 1
34 ; while k <> j do with w[k] do
35 if (xl <= x) and (x <= xh) and
36 (yl <= y) and (y <= yh) then j := k

A Monolithic Solution (using an array)

37 else k := k + 1
38 { k = j, hence if k = n+1 then not found,
39 otherwise w[k] is first window containing (x,y) }
40

41 { Output the result }
42 ; if k = n + 1 then WriteLn(’Desktop’)
43 else WriteLn(k)
44 end { for c }
45

46 end { for i }
47 end.

Quality Assessment (Red needs improvement)

• Comments : at top, variable declarations, statement groups

• Variables : sensible names, single purpose

• Layout : indentation, whitespace, empty lines; begin ... end

• Constants : explicitly named

• Protection : check input for validity

• Modular structure : explicit

Why Modular Structure?

• Correct by design

• Construction by a team

• Verification

• Adaptation

• Reuse

How to avoid, detect, locate, repair errors . . .

A Monolithic Solution (adapted to use a linked list)

1 program Windows2001m2;
2 { Monolithic version using a linked list to store the windows }
3

4 type (*ADDED*)
5 NodeP = ˆNode; { pointer to a node } (*ADDED*)
6 Node = record { node in linked list of windows } (*ADDED*)
7 xl, yl, xh, yh: Integer; { coordinates of window } (*ADDED*)
8 tail: NodeP; { pointer to next window, if not nil } (*ADDED*)
9 end; (*ADDED*)

10

11 var
12 r: Integer; { number of runs (input) }
13 i: Integer; { to count off the runs }
14 n: Integer; { number of windows (input) }
15 w: NodeP; { list of windows (input) } (*CHANGED*)
16 u, v: NodeP; { to construct and traverse list w } (*ADDED*)
17 j: Integer; { to count off the windows }
18 m: Integer; { number of mouse clicks (input) }

A Monolithic Solution (adapted to use a linked list)

19 c: Integer; { to count off the mouse clicks }
20 x, y: Integer; { coordinates of a mouse click (input) }
21 k: Integer; { window number (output) }
22

23 begin
24 ReadLn(r) { number of runs }
25 ; for i := 1 to r do begin
26

27 { Read and store the windows }
28 ReadLn(n) { number of windows }
29 ; w := nil ; u := nil (*ADDED*)
30 { inv: uˆ is last window in list w, if u <> nil } (*ADDED*)
31 ; for j := 1 to n do begin (*CHANGED*)
32 New(v) { create new node } (*ADDED*)
33 ; with vˆ do begin (*ADDED*)
34 ReadLn(xl, yl, xh, yh) { coordinates of window }
35 ; tail := nil (*ADDED*)
36 end { with vˆ } (*ADDED*)

A Monolithic Solution (adapted to use a linked list)

37 ; if u = nil then w := v else uˆ.tail := v (*ADDED*)
38 ; u := v ; v := nil (*ADDED*)
39 end { for j } (*ADDED*)
40

41 { Read and process the mouse clicks }
42 ; ReadLn(m) { number of mouse clicks }
43 ; for c := 1 to m do begin
44 ReadLn(x, y) { coordinates of the mouse click }
45

46 { Find first window that contains (x,y)
47 using a Bounded Linear Search }
48 ; u := w ; v := nil ; k := 1 (*CHANGED*)
49 ; while u <> v do with uˆ do (*CHANGED*)
50 if (xl <= x) and (x <= xh) and
51 (yl <= y) and (y <= yh) then v := u (*CHANGED*)
52 else begin u := tail ; k := k + 1 end (*CHANGED*)
53 { u = v, hence if u = nil then not found, } (*CHANGED*)
54 { otherwise uˆ is first window containing (x,y) }(*CHANGED*)

A Monolithic Solution (adapted to use a linked list)

55

56 { Output the result }
57 ; if u = nil then WriteLn(’Desktop’) (*CHANGED*)
58 else WriteLn(k)
59 end { for c }
60

61 { Deallocate all windows } (*ADDED*)
62 ; while w <> nil do begin (*ADDED*)
63 v := wˆ.tail ; Dispose(w) ; w := v (*ADDED*)
64 end { while } (*ADDED*)
65

66 end { for i }
67 end.

Linked List with Pointers: Recursive Definition

NodeP = ˆNode; { pointer to a node } (*ADDED*)
Node = record { node in linked list of windows } (*ADDED*)

xl, yl, xh, yh: Integer; { coordinates of window } (*ADDED*)
tail: NodeP; { pointer to next window, if not nil } (*ADDED*)
end; (*ADDED*)

A Solution Structured with Routines

1 program Windows2001s1;
2 { Mildly structured version using an array to store the windows }
3

4 const
5 MaxWindowListLength = 10000; { maximum length of a WindowList }
6

7 type
8 Window = record
9 xl, yl, xh, yh: Integer; { lower-left and upper-right corners }

10 { invariant: xl <= xh /\ yl <= yh }
11 end;
12

13 WindowList = record
14 length: 0..MaxWindowListLength; { number of windows in the list }
15 item: array [1..MaxWindowListLength] of Window;
16 { item[1..length] = list of windows }
17 end;
18

A Solution Structured with Routines

19 function InWindow(const w: Window; x, y: Integer): Boolean;
20 { pre: true
21 ret: whether w contains point (x, y) }
22 begin
23 with w do begin
24 Result := (xl <= x) and (x <= xh) and
25 (yl <= y) and (y <= yh)
26 end { with w }
27 end; { InWindow }
28

29 procedure ReadWindowList(out aWindowList: WindowList);
30 { pre: input contains a properly formatted list of windows
31 post: aWindowList has been read from input }
32 { N.B.: uses global variable input }
33 var
34 iWindow: Integer; { to count off the windows }
35 begin
36 with aWindowList do begin

A Solution Structured with Routines

37 ReadLn(length)
38

39 ; for iWindow := 1 to length do begin
40 with item[iWindow] do begin
41 ReadLn(xl, yl, xh, yh)
42 end { with }
43 end { for iWindow }
44

45 end { with aWindowList }
46 end; { ReadWindowList }
47

48 function FindFirstWindow(const aWindowList: WindowList;
49 x, y: Integer): Integer;
50 { pre: true
51 ret: k such that
52 1 <= k <= length + 1,
53 for all i with 1 <= i < k: (x,y) not in item[i], and
54 if k <= length then item[k] contains (x, y)

A Solution Structured with Routines

55 where length and item[_] are from aWindowList }
56 var
57 k, u: Integer; { to traverse aWindowList }
58 begin
59 with aWindowList do begin
60 { Bounded Linear Search }
61 k := 1 ; u := length + 1
62

63 ; while k <> u do begin
64 if InWindow(item[k], x, y) then u := k
65 else Inc(k)
66 end { while }
67 { k = u, hence if k = length+1 then not found,
68 otherwise item[k] contains (x,y) }
69

70 end { with aWindowList }
71 ; Result := k
72 end; { FindFirstWindow }

A Solution Structured with Routines

73

74 procedure ProcessAllMouseClicks(const aWindowList: WindowList);
75 { pre: input contains a properly formatted list of mouse clicks
76 post: the mouse clicks have been read from input and
77 the corresponding output has been written }
78 { N.B.: uses global variables input and output }
79 var
80 nMouseClicks: Integer; { number of mouse clicks (input) }
81 iMouseClick: Integer; { to count off the mouse clicks }
82 x, y: Integer; { coordinates of a mouse click (input) }
83 k: Integer; { window number (output) }
84 begin
85 ReadLn(nMouseClicks)
86

87 ; for iMouseClick := 1 to nMouseClicks do begin
88 ReadLn(x, y)
89 ; k := FindFirstWindow(aWindowList, x, y)
90 ; if k <= aWindowList.length then WriteLn(k)

A Solution Structured with Routines

91 else WriteLn(’Desktop’)
92 end { for iMouseClick }
93

94 end; { ProcessAllMouseClicks }
95

96 procedure ProcessOneRun;
97 { pre: input contains a properly formatted run
98 post: the run has been read from input and
99 the corresponding output has been written }

100 { N.B.: uses global variables input and output }
101 var
102 theWindowList: WindowList; { list of windows (input) }
103 begin
104 ReadWindowList(theWindowList)
105 ; ProcessAllMouseClicks(theWindowList)
106 end; { ProcessOneRun }
107

108 { main program }

A Solution Structured with Routines

109 var
110 nRuns: Integer; { number of runs (input) }
111 iRun: Integer; { to count off the runs }
112

113 begin
114 ReadLn(nRuns)
115

116 ; for iRun := 1 to nRuns do begin
117 ProcessOneRun
118 end { for iRun }
119

120 end.

Parameter Kinds in Pascal

function InWindow(const w: Window; x, y: Integer): Boolean;

procedure ReadWindowList(out aWindowList: WindowList);

• value parameter : local variable initialized (copied) at call time

• out parameter : body can set variable supplied at call time

• var parameter : body can update variable supplied at call time

• const parameter : no copy; body cannot change it

Divide and Conquer through Contracts

procedure ReadWindowList(out aWindowList: WindowList);
{ pre: input contains a properly formatted list of windows

post: aWindowList has been read from input }
{ N.B.: uses global variable input }

Contract

Precondition Postcondition

User obligation benefit

Party ↓ ↑

Maker benefit → obligation

The Contract as the (Only) Interface

function InWindow(const w: Window; x, y: Integer): Boolean;
{ pre: true

ret: whether w contains point (x, y) }

begin
with w do begin

Result := (xl <= x) and (x <= xh) and
(yl <= y) and (y <= yh)

end { with w }
end; { InWindow }

if InWindow(item[k], x, y) then u := k

Contract

↗ ↖
Call Body

Software Architecture: Key Ingredients

• Building blocks (the units of design, construction, verification,
adaptation, reuse)

– Variables and statements

– Routines (functions, procedures, with parameters)

– Data types (classes, with methods)

– Object Pascal ‘units’, components, packages, . . .

• Relationships between building blocks (dependencies, coupling)

Software Architecture: Static Call Graph

program Windows2001s1

!

procedure ProcessOneRun
"

"
"#

procedure ReadWindowList

$
$

$%

procedure ProcessAllMouseClicks

!

function FindFirstWindow

!

function InWindow

Local versus global variables

Alternative Architecture: Decomposition with Points

program Windows2001s2

!

procedure ProcessOneRun
"

"
"#

procedure ReadWindowList

$
$

$%

procedure ProcessAllMouseClicks

!

function FindFirstWindow

!

function InWindow

!

procedure ReadWindow

!

procedure ReadPoint

&
&

&
&

&
&

&
&&'

!

function PointDominated

Architecture on the Basis of Data Types

WindowList

0..1

0..*

{ordered}

Window

1 1

LowerLeft UpperRight

Point

Interface styles

• Interface coordinates calls of routines: ‘control flow’

• Console Interface : Program-driven control flow

Interactive (via prompting)

batch (via command-line arguments and files)

• Graphical User Interface (GUI): Event-driven control flow

Events: Keyboard, mouse, buttons, menus, sliders, . . .

‘Main event loop’ : wait for ‘event’ and call its handler (routine)

GUI building blocks

Lazarus Component Library (LCL):

• Windows, . . .

• Buttons, . . .

• Text areas, . . .

• Menus, . . .

• Graphical objects, . . .

Object-oriented programming

• Class : a collection of data items and associated operations

type for a variabele, compare to Integer

fields, methods, properties

• Object : an instance of a class

value of a variabele, compare to a concrete number, like 5

• Hierarchy : relationships bedtween classes via inheritance

TComponent ←− TControl ←− . . . ←− TButton

Constructing GUI programs

Programming versus Configuring

Visual ‘Programming’: construct a program by ‘clicking it together’

Convenient Facilities

• procedure ShowMessage ( const Msg: String );

Shows dialog window with Msg; execution waits for OK-click

• function IntToStr ( Value: Integer ): String;

Converts Value to a String

• function StrToInt ( const s: String ): Integer ;

Converts s to an Integer

• function TryStrToInt ( const s: String; out i: Integer ): Boolean;

• function Val ( const s: String; out V ; out Code: Word );

Convenient Facilities

• procedure Close ; { in Form }

Closes the form (in main form: terminates program execution)

• To allow a menu on a Form : BorderStyle := bsSizeable

Files in a Lazarus GUI program

What you should save and submit of a Lazarus GUI program:

• *.lpi : Lazarus Project Information (XML)

• *.lpr : Lazarus Program (Pascal code)

• *.pas : Lazarus Unit (Pascal code)

• *.lfm : Lazarus Form (Object configuration data in text)

• *.lrs : Lazarus Resources (binary data)

Other files contain only secundary information

What Lies Ahead

• Abstract Data Types and object-oriented programming

• Dynamic variables and pointers

• Recursion , both in control and in data

• Contract checking through Assert

• Event-driven, interactive Graphical User Interfaces

