
Today’s Topics

• Branching dynamic data structures with pointers:

– Trees

– Binary Trees (BTs)

– Binary Search Trees (BSTs)

• More recursion

Linear Structures: Linked Lists

• Storage overhead : 1 or 2 pointers per cell

Single-linked versus double-linked (bi-directional)

Double linking simplifies insertion and deletion of nodes.

• Access/modification time : worst-case proportional to list length

Linear in the actual size: O(N)

How to improve the worst-case time for operations?

Branching Structures: (Binary (Search)) Trees

In computing, trees are everywhere

• Syntax trees

–• Directory trees

• Widget trees

• Search trees

• . . .

Tree Concepts and Terminology

• Graph: node, edge (connected, no cycles)

• Parent, child, subtree, order on children

• Root, leaf, front, root path

• Arity (fixed, variable)

• Height, balance

• Traversal: pre-order, in-order, post-order

c© 2008, T. Verhoeff @ TUE.NL 5 Programming – Block C: Lecture 15

Binary Trees

The set BT(S) of Binary Trees over a set S is inductively defined by

(Basis) ε ∈ BT(S) (the empty tree)

(Step) if L ∈ BT(S), a ∈ S, R ∈ BT(S), then 〈L, a, R〉 ∈ BT(S)

Some examples in BT(N): 7

• ε

• 〈ε,2, ε〉
• 〈〈ε,2, ε〉,3, 〈ε,5, ε〉〉
• 〈〈〈ε,2, ε〉,3, 〈ε,5, ε〉〉,7, 〈ε,11, ε〉〉

Recursive Functions on Binary Trees

The height h.t of a binary tree t is recursively defined by

(Basis) h.ε = 0

(Step) h.〈L, a, R〉 = 1 + max(h.L, h.R)

Examples:

• h.〈ε,11, ε〉 = 1

• h.〈ε,2, ε〉,3, 〈ε,5, ε〉〉 = 2

• h.〈〈〈ε,2, ε〉,3, 〈ε,5, ε〉〉,7, 〈ε,11, ε〉〉 = 3

Binary Tree Representation in Pascal

1 type
2 TData = ...; // type for content of the binary trees
3

4 PNode = ˆTNode;
5

6 TNode = // node of binary tree over TData
7 record
8 FData: TData; // value stored in this node
9 FLeft: PNode; // left subtree

10 FRight: PNode; // right subtree
11 end;
12

13 TBTData = PNode; // Binary Trees over TData
14 // data invariants
15 // I0: all root paths via FLeft/FRight terminate in nil
16 // I1: if two root paths both reach u then u = nil
17 // Abstraction function [[t]] for t: TBTData equals
18 // empty tree if t=nil else <[[tˆ.FLeft]], tˆ.FData, [[tˆ.FRight]]>

Binary Tree Type in Pascal: Examples

t is OK t0 violates I0 t1 violates I1

Binary Trees in Pascal: Observations

• Storage overhead : 2 pointers per node

• # nil pointers = # nodes + 1

• Maximum # nodes in tree of height N = 2N − 1

• Minimum # nodes in tree of height N = N

• log2(1 + #nodes) ≤ height ≤ #nodes

• In a balanced binary tree, the root path length to arbitrary node
is potentially logarithmic in the total number of nodes.

Creating a Binary Tree in Pascal

The empty tree ε is represented by nil .

A nonempty tree is best created through an auxiliary function:

1 function MakeBTData(a: TData; L, R: TBTData): TBTData;
2 // pre: L and R do not share any nodes
3 // ret: <L, a, R>
4 begin
5 New(Result);
6 with Resultˆ do begin
7 FData := a;
8 FLeft := L;
9 FRight := R;

10 end; { with }
11 // Result satisfies I0 and I1
12 end;

Creating a Binary Tree in Pascal: Example

1 var
2 t: TBTData; // a binary tree over TData

3

4 begin
5 t := MakeBTData(7,

6 MakeBTData(3,

7 MakeBTData(2, nil, nil),
8 MakeBTData(5, nil, nil)),
9 MakeBTData(11, nil, nil));

Height of Binary Tree in Pascal

1 uses
2 Math; // for Max
3

4 function Height(t: TBTData): Integer;
5 // pre: true
6 // ret: h.t
7 begin
8 if t = nil then begin
9 Result := 0;

10 end
11 else { t <> nil } begin
12 Result := 1 + Max ( Height(tˆ.FLeft), Height(tˆ.FRight) );
13 // termination guaranteed because of I0
14 end;
15 end;

Binary Tree Traversals

Standard orders for “visiting” the nodes of a binary tree:

• Pre-order : first the node itself, then the subtrees

• In-order : first left subtree, next the node, then right subtree

• Post-order : first the subtrees, then the node

Coding techniques: recursively or with repetition and self-made stack

Pre-order Traversal in Pascal (recursive)

1 procedure PreOrder(t: TBTData);
2 // pre: true
3 // post: all nodes of t have been processed in pre-order
4 begin
5 if t = nil then begin
6 // skip
7 end
8 else { t <> nil } begin
9 with tˆ do begin

10 ... process FData ...;
11 PreOrder(FLeft);
12 PreOrder(FRight);
13 end; { with }
14 end;
15 end;

In-order Traversal in Pascal (recursive)

Slightly “optimized” code

1 procedure InOrder(t: TBTData);

2 // pre: true

3 // post: all nodes of t have been processed in in-order

4 begin
5 if t <> nil then begin
6 with tˆ do begin
7 InOrder(FLeft);

8 ... process FData ...;

9 InOrder(FRight);

10 end; { with }

11 end; { if }

12 end;

c© 2008, T. Verhoeff @ TUE.NL 16 Programming – Block C: Lecture 15



Parameterized Post-order Traversal in Pascal (recursive)

1 type TDataAction = procedure ( const AData: TData );
2

3 procedure PostOrder(t: TBTData; ADataAction: TDataAction);
4 // pre: true
5 // post: all nodes of t have been processed in post-order
6 // by applying ADataAction
7 begin
8 if t <> nil then begin
9 with tˆ do begin

10 PostOrder(FLeft, ADataAction);
11 PostOrder(FRight, ADataAction);
12 ADataAction(FData);
13 end; { with }
14 end; { if }
15 end;

Apply Parameterized Post-order Traversal in Pascal

1 procedure WriteData(const AData: TData);

2 begin
3 Writeln(... AData ...)

4 end;
5

6 var
7 t: TBTData; // a binary tree

8

9 // WriteData has type TDataAction

10 PostOrder(t, WriteData);

Apply Post-order Traversal in Pascal (2)

1 var
2 count: Integer; // global counter for CountData
3

4 procedure CountData(const AData: TData {; globvar count});
5 // effect: count := count + 1
6 begin
7 count := count + 1
8 end;
9

10 var
11 t: TBTData; // a binary tree over TData
12

13 count := 0;
14 PostOrder(t, CountData);
15 // count = # nodes in t

Binary Search Trees

Time to find data in a binary tree is still worst-case linear in the size

Can be improved by keeping data sorted

The list ".t of a binary tree t is inductively defined by

(Basis) ".ε = ε (the empty list)

(Step) ".〈L, a, R〉 = ".L ++ [a] ++ ".R (cf. in-order traversal)

A Binary Search Tree is a binary tree t with additional invariant:

• ".t is strictly increasing (w.r.t. a suitable order on S)

Binary Search Trees in Pascal

1 type
2 TBSTData = TBTData; // Binary Search Tree over TData

3 // data invariant

4 // I2: in-order list is strictly increasing

5 // for every subtree <L, a, R>: data in L < a < data in R

6 TCompareData = ( ls, eq, gt ); // less, equal, greater

7

8 function CompareData(X, Y: TData): TCompareData;

9 begin
10 if ‘‘X < Y’’ then Result := ls

11 else if ‘‘X = Y’’ then Result := eq

12 else { ‘‘X > Y’’ } Result := gt

13 end;

Find Data in Binary Search Tree (Recursive)

1 function FindRec(const X: TData; t: TBSTData): PNode;
2 // pre: true
3 // ret: r with t -*-> r /\ rˆ.FData = X if X in t else nil
4 begin
5 if t = nil then Result := nil
6 else { t <> nil } begin
7 with tˆ do begin
8 case CompareData(X, FData) of
9 ls: Result := FindRec(X, FLeft);

10 eq: Result := t;
11 gt: Result := FindRec(X, FRight);
12 end; { case }
13 end; { with }
14 end; { else }
15 end;

Find Data in Binary Search Tree (Iterative)

1 function FindIter(const X: TData; t: TBSTData): PNode;
2 // pre: t = T
3 // ret: r with t -*-> r /\ rˆ.FData = X if X in T else nil
4 begin
5 Result := nil;
6 // inv: T -*-> t -*-> Result /\ X in T == X in t /\
7 // (Result <> nil ==> Resultˆ.FData = X)
8 while t <> Result do { t <> nil } with tˆ begin
9 case CompareData(X, FData) of

10 ls: t := FLeft;
11 eq: Result := t;
12 gt: t := FRight;
13 end; { case }
14 end; { while/with }
15 end;

Insert Data into Binary Search Tree (Recursive)

1 procedure Insert(const X: TData; var t: TBSTData);
2 // pre: true
3 // post: t = initial t extended with X
4 begin
5 if t = nil then t := MakeBTData(X, nil, nil)
6 else { t <> nil } begin
7 with tˆ do begin
8 case CompareData(X, FData) of
9 ls: Insert(X, FLeft);

10 eq: { skip }; // already present, do not duplicate
11 gt: Insert(X, FRight);
12 end; { case }
13 end; { with }
14 end; { else }
15 end;

Loose Ends

• Minimum, Maximum

• Iterative traversals, iterative insert

• Delete item, dispose entire tree

• Merge (union)

• Maintaining balance

• Sorting algorithms

What Lies Ahead

Course Code Year Blocks

Programming Methods 2IP15 1 D-E-F

Data Structures 2IL05 1 D-E-F

Software Specification 2IW05 2 A-B-C

Algorithmics 2IL15 2 D-E-F

Distributed Algorithms 2IL25 2 D-E-F

Functional Programming 2IA05 3 D-E

Software Engineering 2IP25 3 D

Software Engineering Project 2IP35 3 D-E-F

Verhaal met moraal

Jorge Luis Borges

Argentijns schrijver en dichter

1899 – 1986

LA ROSA DE PARACELSO

—

De roos van Paracelsus

